@ECHO OFF
REM Example of how to run the RESTful DDS service
REM Usage: startRESTfulDDS.bat [wwwRootURI [space separated user-jarlist]]
IF {%1}=={} (
  SET URI="file:///%CD%\www"
) ELSE (
  SET URI=%1
  SHIFT
)

REM RESTfulDDS jar
SET CP=lib\org.opensplice.restful.service.jar

REM External jars
REM Fill variables so they point to external jar files
CALL externals.bat
SET CP=%CP%;"%DDS_JAR%";%GSON_JAR%;%RESTLET_IMPL_JAR%;%RESTLET_INTF_JAR%

REM Userdefined jars
:Loop
IF {%1}=={} GOTO :Done
SET CP=%CP%;%1
SHIFT
GOTO :Loop
:Done

START java -cp %CP% org.opensplice.restful.service.RESTfulDDSMain %URI%